
Change 1 – New constant tsc_NoOfIMSPDNs
	  const integer tsc_NoOfIMSPDNs := 2;

	  type component EUTRA_PTC {
    var  EUTRA_Global_Type vc_EUTRA_Global;

    port EUTRA_SYSTEM_PORT          SYS;

    port EUTRA_SYSIND_PORT          SYSIND;

    port EUTRA_VNG_PORT             VNG;

    port CDMA2000_TUNNELLING_PORT   C2KTUNNEL;
    port EUTRA_SRB_PORT             SRB;

    port EUTRA_NB_NASCTRL_PORT      NASCTRL;

    port EUTRA_DRB_PORT             DRB;

    port EUTRA_MRB_PORT             MRB;

    port UT_PTC_MTC_PORT            UT;

    port IRAT_CO_ORD_PORT           UTRAN;

    port IRAT_CO_ORD_PORT           GERAN;

    port IRAT_CO_ORD_PORT           CDMA2000;

    port IRAT_CO_ORD_PORT           WLAN;

    port IRAT_CO_ORD_PORT           NBIOT;

    port IP_RAT_CTRL_PORT           IP;

    port IMS_IPCAN_CO_ORD_PORT      IMS[tsc_NoOfIMSPDNs];

  };

	  type component GERAN_PTC {

    var  GERAN_Global_Type vc_GERAN_Global;

    timer tc_TBF;

    port UT_PTC_MTC_PORT        UT;

    port IRAT_CO_ORD_PORT       EUTRA;

    port IRAT_CO_ORD_PORT       UTRAN;

    port GERAN_CL1_PORT         G_CL1;

    port GERAN_CLLC_PORT        G_CLLC;

    port GERAN_CRLC_PORT        G_CRLC;

    port GERAN_CSNDCP_PORT      G_CSNDCP;

    port GERAN_L2_PORT          G_L2;

    port GERAN_LLC_PORT         G_LLC;

    port GERAN_RLC_PORT         G_RLC;

    port GERAN_SNDCP_PORT       G_SN;

    port IP_RAT_CTRL_PORT       IP;

    port IMS_IPCAN_CO_ORD_PORT  IMS[tsc_NoOfIMSPDNs];

  };

	  type component NR5GC_PTC extends NR_BASE_PTC {

    var NR5GC_Global_Type           vc_NR5GC_Global;

    port IMS_IPCAN_CO_ORD_PORT      IMS[tsc_NoOfIMSPDNs];

    port NG_NASCTRL_PORT            NASCTRL;

  };

	  type component UTRAN_IRAT_PTC extends UTRAN_IP_PTC {
    port IRAT_CO_ORD_PORT       EUTRA;

    port IRAT_CO_ORD_PORT       GERAN;

    port IMS_IPCAN_CO_ORD_PORT  IMS[tsc_NoOfIMSPDNs];
  };

	  type component WLAN_PTC {

    var  WLAN_Global_Type          vc_WLAN_Global;

    port UT_PTC_MTC_PORT           UT;

    port IRAT_CO_ORD_PORT          EUTRA;

    port IRAT_CO_ORD_PORT          UTRAN;

    port IP_RAT_CTRL_PORT          IP;

    port IMS_IPCAN_CO_ORD_PORT     IMS[tsc_NoOfIMSPDNs];

    port WLAN_AP_SYSTEM_PORT       SYS;

    port WLAN_AP_SYSCMD_IND_PORT   SYS_CMDIND;

    port WLAN_ePDG_SYSTEM_PORT     SYS_EPDG;

    port WLAN_ePDG_SYSCMD_IND_PORT SYS_EPDG_CMDIND;

  };

	  type record length(tsc_NoOfIMSPDNs) of IMS_Server_Type IMS_ServerArray_Type;

	  type component IP_PTC {

    var IP_PTC_DRB_RoutingInfo_Type     vc_IP_PTC_DRB_RoutingInfo;

    var PDN_IP_AddrServer_Type          vc_IP_PTC_PDN_AddrServer;

    var IP_PTC_IMS_Info_Type            vc_IP_PTC_IMS_Info;

    var PDN_MappingTable_Type           vc_IP_PTC_PDN_MappingTable;

    var IP_PTC_WLAN_Info_Type           vc_IP_PTC_WLAN_Info;

    // ports connected to IMS:

    port IP_IMS_CTRL_PORT       IMS_CTRL[tsc_NoOfIMSPDNs];

    port IP_IMS_SERVER_PORT     IMS_Server[tsc_NoOfIMSPDNs];

    port IP_IMS_CLIENT_PORT     IMS_Client[tsc_NoOfIMSPDNs];

    // coordination ports:

    port IP_PTC_CTRL_PORT       EUTRA_CTRL;

    port IP_PTC_CTRL_PORT       UTRAN_CTRL;

    port IP_PTC_CTRL_PORT       GERAN_CTRL;

    port IP_PTC_CTRL_PORT       WLAN_CTRL;

    port IP_PTC_CTRL_PORT       IPCAN_CTRL;

    port IP_PTC_CTRL_PORT       NR_CTRL;

    // coordination ports to HTTP_PTC

    port IP_HTTP_CTRL_PORT      HTTPCTRL;

    port IP_HTTP_DATA_PORT      HTTPDATA;

    // coordination port for sending of UDP packets to the UE (e.g. for OMA Push)

    port IP_UDP_CLIENT_PORT     UDPCLIENT;

    // ports to IP_MUX_PTC

    port IP_SOCKET_PORT         IP_SOCK_DATA;

    port IP_SOCKET_PORT         IP_SOCK_CTRL;

    // system ports

    port IP_CONTROL_PORT        IP_CTRL;

    port IPSEC_CONTROL_PORT     IPSEC_CTRL;

  };


Change 2 – New Function f_PDN_PdnIndex2ImsIndex
	  function f_PDN_PdnIndex2ImsIndex(PDN_Index_Type p_PdnIndex) return integer

  {

    var integer v_Index := f_PDN_PdnIndex2Integer(p_PdnIndex);

    if (v_Index >= tsc_NoOfIMSPDNs) {

      FatalError(__FILE__, __LINE__, "Invalid PDN index");

    }

    return v_Index;

  }

	  function f_EUTRA_ResetIMS(PDN_Index_Type p_PdnIndex) runs on EUTRA_PTC

  {

    var integer v_Index := f_PDN_PdnIndex2ImsIndex(p_PdnIndex);

    f_IMS_IPCAN_SendCoOrdMsg(IMS[v_Index], cms_IPCAN_IMS_Reset);

  }

	  function fl_SPI_GetValidValue(integer p_Offset) runs on IMS_PTC return integer

  {

    var integer v_Index := f_PDN_PdnIndex2ImsIndex(vc_IMS_Global.PdnIndex);

    var IntegerList_Type v_ValidValueList := vc_IMS_Global.ImsInfo.SPI_ValidValueList[v_Index];

    var integer v_ValidValueListLength := lengthof(v_ValidValueList);

    var integer v_SPI;

    if (v_ValidValueListLength > 0) {

      if (v_ValidValueListLength <= p_Offset) {

        FatalError(__FILE__, __LINE__, "Not enough SPI values");

      }

      v_SPI := v_ValidValueList[p_Offset];

    } else {

      v_SPI := 256 + (v_Index * 1024) + p_Offset; /* @sic R5s130651: v_Index used to get different SPIs for different PDNs sic@ */

    }

    if (v_SPI < 256) {  /* According to RFC 2406, SPI values from 0 to 255 are reserved */

      FatalError(__FILE__, __LINE__, "Invalid SPI value");

    }

    return v_SPI;

  }

	  function fl_IP_ImsServer_Start(PDN_Index_Type p_PdnIndex) runs on IP_PTC

  {
    var integer v_IndexVal := f_PDN_PdnIndex2ImsIndex(p_PdnIndex);

    var IMS_Server_Type v_ImsServer := vc_IP_PTC_IMS_Info.Server[v_IndexVal];

    var PDN_AddressInfo_Type v_PDN_AddressInfo := f_PDN_AddressInfo_Get(p_PdnIndex);

    v_ImsServer := f_IP_IMS_Stop(v_ImsServer);

    vc_IP_PTC_IMS_Info.Server[v_IndexVal] := f_IP_IMS_Start(v_ImsServer,

                                                            v_PDN_AddressInfo.PCSCF_IPAddressIPv4,

                                                            v_PDN_AddressInfo.PCSCF_IPAddressIPv6,

                                                            p_PdnIndex);

  }

	  function f_NR5GC_SwitchOnAndResetIMS() runs on NR5GC_PTC

  {

    f_UT_SwitchOnUE(UT);

    f_IMS_IPCAN_SendCoOrdMsg(IMS[f_PDN_PdnIndex2ImsIndex(PDN_1)], cms_IPCAN_IMS_Reset);

  }


Change 3 – Further changes in IP_PTC_Main
	  function fl_IP_PTC_Init() runs on IP_PTC

  {

    var integer i;

    vc_IP_PTC_DRB_RoutingInfo := {

      Table := {},

      RawMode := false

    };

    vc_IP_PTC_WLAN_Info := valueof(cs_IP_PTC_WLAN_Info_Init);  /* @sic R5w160107 slide 7 sic@ */

    for (i:=0; i < tsc_NoOfPDNs; i:=i+1) {

      vc_IP_PTC_PDN_AddrServer.DHCP[i]        := valueof(cs_IP_AddrServer_Init);

      vc_IP_PTC_PDN_AddrServer.DNS_UdpIPv4[i] := valueof(cs_IP_AddrServer_Init);

      vc_IP_PTC_PDN_AddrServer.DNS_UdpIPv6[i] := valueof(cs_IP_AddrServer_Init);

      vc_IP_PTC_PDN_AddrServer.ICMPv6[i]      := valueof(cs_IP_AddrServer_Init);


      vc_IP_PTC_PDN_MappingTable[i].DefaultDrb := omit;  /* @sic R5s150648 Change 2.1 sic@ */

    }
    for (i:=0; i < tsc_NoOfIMSPDNs; i:=i+1) {

      vc_IP_PTC_IMS_Info.Server[i]            := valueof(cs_IMS_Server_Init);

    }
  }

	  function fl_StopPDNs(template (omit) PDN_Index_Type p_PdnIndex := omit) runs on IP_PTC

  {




    var integer i;

    // Reset address servers and mapping info 

    for (i := 0; i < tsc_NoOfPDNs; i := i + 1) {

      if (not isvalue(p_PdnIndex) or (i == f_PDN_PdnIndex2Integer(valueof(p_PdnIndex)))) 

        fl_IP_AddrServers_Stop(i);

        vc_IP_PTC_PDN_MappingTable[i].DefaultDrb := omit;   /* @sic R5s150648 Change 6 sic@ */

      }

    }

    // Reset IMS server(s) 

    for (i := 0; i < tsc_NoOfIMSPDNs; i := i + 1) {

      if (not isvalue(p_PdnIndex) or (vc_IP_PTC_IMS_Info.Server[i].PdnIndex == valueof(p_PdnIndex))) {   // NOTE: requires that the same PdnIndex is used as for start of the IMS server

        vc_IP_PTC_IMS_Info.Server[i] := f_IP_IMS_Stop(vc_IP_PTC_IMS_Info.Server[i]); 

      }

    }
  }

	  altstep a_ApplicationCtrl(IP_PTC_CTRL_PORT p_CtrlPort) runs on IP_PTC

  {

    var IP_ConfigReq_Type v_IP_ConfigReq;

    var PDN_Index_Type v_PdnIndex;

    var IP_DrbInfo_Type v_DrbInfo;

    var IP_DrbInfo_Type v_CurrentDrbInfo;

    var EUTRA_CellId_Type v_EutraCellId;

    var integer i;

    [] p_CtrlPort.receive(cr_IP_CtrlAnyReq) -> value v_IP_ConfigReq {

      if (ischosen(v_IP_ConfigReq.StartPDN)) {

        v_PdnIndex := v_IP_ConfigReq.StartPDN.PdnId;

        v_DrbInfo := v_IP_ConfigReq.StartPDN.DrbInfo;

        f_IP_PTC_PDN_SetDrbInfoForDefaultDrb(v_PdnIndex, v_DrbInfo);

        select (v_PdnIndex) {

          case (PDN_1, PDN_2, PDN_2a) {

            fl_IP_AddrServers_Start(v_PdnIndex, v_DrbInfo);

            fl_IP_ImsServer_Start(v_PdnIndex);

          }

          case (PDN_Internet, nrPDN_Internet, PDN_3) { // @sic R5s191074 sic@

            fl_IP_AddrServers_Start(v_PdnIndex, v_DrbInfo);

          }

          case (ePDG_IMS1, ePDG_IMS2) {

            fl_IP_ImsServer_Start(v_PdnIndex);

          }

          case (ePDG_XCAP) {

            fl_IP_AddrServers_Start(v_PdnIndex, v_DrbInfo, dnsServerOnly);

          }

          case (ePDG_Internet) {

            // nothing to be done

          }

        }

      } else if (ischosen(v_IP_ConfigReq.StopPDN)) {

        if (ischosen(v_IP_ConfigReq.StopPDN.PdnId)) {       /* @sic R5s131000: stop particular PDN sic@ */



          v_PdnIndex := v_IP_ConfigReq.StopPDN.PdnId;

          fl_StopPDNs(v_PdnIndex);
        } else {




          fl_StopPDNs();
        }

      } else if (ischosen(v_IP_ConfigReq.StartWLAN)) {

        fl_WLAN_Start(v_IP_ConfigReq.StartWLAN.AddressInfo, v_IP_ConfigReq.StartWLAN.DrbInfo, v_IP_ConfigReq.StartWLAN.ExplicitDnsHandling);

      } else if (ischosen(v_IP_ConfigReq.StopWLAN)) {

        fl_WLAN_Stop();

      } else if (ischosen(v_IP_ConfigReq.ChangeEutraCell)) {   /* @sic R5s150648 Change 6 sic@ */

        v_EutraCellId := v_IP_ConfigReq.ChangeEutraCell;

        if (f_RoutingTable_ChangeEutraCell(v_EutraCellId)) {

          for (i:=0; i<tsc_NoOfPDNs; i:=i+1) {

            f_IP_DrbInfo_ChangeEutraCell(vc_IP_PTC_PDN_MappingTable[i].DefaultDrb, v_EutraCellId);

          }

        }

      } else if (ischosen(v_IP_ConfigReq.ChangeDrbInfo)) {   /* @sic R5s150648 Change 6 sic@ */

        v_PdnIndex := v_IP_ConfigReq.ChangeDrbInfo.PdnId;

        v_DrbInfo := v_IP_ConfigReq.ChangeDrbInfo.DrbInfo;

        v_CurrentDrbInfo := f_IP_PTC_PDN_GetDrbInfoForDefaultDrb(v_PdnIndex);

        if (f_RoutingTable_ChangeDrbInfo(v_CurrentDrbInfo, v_DrbInfo)) {

          f_IP_PTC_PDN_SetDrbInfoForDefaultDrb(v_PdnIndex, v_DrbInfo);

        }

      } else if (ischosen(v_IP_ConfigReq.SendRouterAdvertisement)) {

        fl_ICMPv6_SendRouterAdvertisement(v_IP_ConfigReq.SendRouterAdvertisement.PdnId);

      } else if (ischosen(v_IP_ConfigReq.ExplicitRouting)) {

        if (ischosen(v_IP_ConfigReq.ExplicitRouting.Add)) {

          f_IP_PTC_RoutingTable_AddEntry(v_IP_ConfigReq.ExplicitRouting.Add.IpInfo,

                                         v_IP_ConfigReq.ExplicitRouting.Add.DRB,

                                         v_IP_ConfigReq.ExplicitRouting.Add.DataMode);

        } else if (ischosen(v_IP_ConfigReq.ExplicitRouting.Remove)) {  /* @sic R5s160131 sic@ */

          f_IP_PTC_RoutingTable_RemoveEntry(v_IP_ConfigReq.ExplicitRouting.Remove);

        }

      } else if (ischosen(v_IP_ConfigReq.ChangeToRawmode)) {  /* @sic R5s150074 - Additional change sic@ */

        f_RoutingInfo_ChangeMode(true);   // @sic R5s200044 change 4, R5s200490 sic@

      } else if (ischosen(v_IP_ConfigReq.RestoreRoutingTable)) { // @sic R5s200044 change 4 sic@

        f_RoutingInfo_ChangeMode(false);  // @sic R5s200490 sic@!

      }

      p_CtrlPort.send(cs_IP_CommonCnf);

    }

  }


Change 4 – tsc_Index_PDN1/2 renamed to tsc_Index_IMS1/2
tsc_Index_PDN1/2 renamed to tsc_Index_IMS1/2 in all TTCN modules.
